
esta aula

uma exploração prática de ferramentas que se sustentam sobre as
linguagens regulares e algébricas, os seus conceitos, e os seus algoritmos

• a biblioteca genlex
• a biblioteca str
• a ferramenta ocamllex
• a ferramenta ocamlyacc
• a ferramenta menhir

análise léxica, uma curta introdução

análise léxica

Ihavetheabilitytoarrange1’sand0’sinsuchanorderthatanx86processorcanactually
interpretandexecutethosecommands.It’scalledComputerProgramming,butit’sthe
closestthatamancanevergettogivingbirthinmyopinion.AndIsomehowfeelresponsi
bleforthefutureexistenceandacceptanceofmy"child".I’dspendhourstryingtofindth
etinybugthatcausesmychildtomisbehaveoractstrangely.Butthat’smymildsuperpo
wer...Imaketheworldabetterplacebywritingmindlessback-endprogramsthatno-on
ewilleverseenorevenknowthatit’sthere.ButIknow;andthat’sallthatmatters.Stadium
airconditioningfails-Fansprotest

análise léxica

I have the ability to arrange 1’s and 0’s in such an order that an x86
processor can actually interpret and execute those commands. It’s called
Computer Programming, but it’s the closest that a man can ever get to
giving birth in my opinion. And I somehow feel responsible for the future
existence and acceptance of my "child". I’d spend hours trying to find the
tiny bug that causes my child to misbehave or act strangely. But that’s my
mild superpower... I make the world a better place by writing mindless
back-end programs that no-one will ever see nor even know that it’s there.
But I know; and that’s all that matters.

atributed to Alucard

Stadium air conditioning fails — Fans protest

anonymous newspaper headline

análise léxica

a análise léxica define-se como o corte do texto fonte em « palavras »

à semelhança do que acontece em linguagem natural, este corte do texto
em palavras facilita o trabalho da fase seguinte de análise do texto, a
análise sintáctica

estas palavras, no contexto da análise léxica, são designadas de lexemas
(tokens)

SMDS TC 7



análise léxica: um exemplo

fonte = sequência de caracteres

fun x -> (* uma função *)
x+1

↓
análise léxica

↓

sequência de lexemas

fun x -> x + 1
...

...
↓

análise sintáctica
↓

sintaxe abstracta

Fun

"x" App

App

Op

+

Var

"x"

Const

1

a biblioteca Genlex

Genlex

Genlex é um módulo OCaml da biblioteca
standard para criar rapidamente analisadores
léxicos para linguagens simples

a documentação está disponível aqui (link)

pressupõe que os lexemas possam todos ser
classificados em 6 categorias:

• palavras chave (Kwd)

• identificadores (Ident)

• números inteiros (Int)

• números flutuantes (Float)

• cadeias de caracteres (String)

• caracteres (Char)

Um exemplo simples

open Genlex
open Stream

let keywords =
(* serve para destacar as palavras chaves dos identificadores*)
[ "LET"; "PRINT"; "IF"; "THEN"; "ELSE"; "+"; "*"; "/"; "=" ]

let line_lexer l = Genlex.make_lexer keywords (Stream.of_string l)

Um exemplo simples

# let tokenstream = line_lexer "LET x = x + y * 3" ;;
val tokenstream : Genlex.token Stream.t = <abstr>
# Stream.next tokenstream;;
- : Genlex.token = Kwd "LET"
# next tokenstream;;
- : Genlex.token = Ident "x"
# next tokenstream;;
- : Genlex.token = Kwd "="
# next tokenstream;;
- : Genlex.token = Ident "x"
# next tokenstream;;
- : Genlex.token = Kwd "+"
# next tokenstream;;
- : Genlex.token = Ident "y"
# next tokenstream;;
- : Genlex.token = Kwd "*"
# next tokenstream;;
- : Genlex.token = Int 3
# next tokenstream;;
Exception: Stream.Failure.

SMDS TC 12



a biblioteca Str

Str

Str é um módulo OCaml da biblioteca standard para criar rapidamente
processadores de texto baseados em expressões regulares

a documentação está disponível aqui (link)

Str

Str é um módulo OCaml da
biblioteca standard para criar
rapidamente processadores de texto
baseados em expressões regulares

a documentação está disponível aqui
(link)

Str

Str é um módulo OCaml da
biblioteca standard para criar
rapidamente processadores de texto
baseados em expressões regulares

a documentação está disponível aqui
(link)

Str

um exemplo simples

(* um pequeno utilitário *)
let (=~) s re = Str.string_match (Str.regexp re) s 0

# let pad = "[ACGT]*TGC";;
val pad : string = "[ACGT]*TGC"
# let st = "ATTGCAGTAGGACTCGCCTGATGCAGTC";;
- : string = "ATTGCAGTAGGACTCGCCTGATGCAGTC"
# st =~ pad;;
- : bool = true
# matched_string st;;
- : string = "ATTGCAGTAGGACTCGCCTGATGC"

outro exemplo

let portuguese_date_of d =
match d with
[mm; dd; aa] -> dd^"/"^mm^"/"^aa

| _ -> failwith "Bad date format"

(* atenção: string cortada para os efeitos da aula *)
let english_date_format = Str.regexp
"\\([0-9][0-9]\\)\\.\\([0-9][0-9]\\)\\.\\(\\([0-9][0-9]\\)
\\|\\([0-9][0-9][0-9][0-9]\\)\\)"

let trans_date l =
try
let _ =Str.search_forward english_date_format l 0 in
let d1 = Str.matched_string l in
let d2 = portuguese_date_of (Str.split (Str.regexp "\.") d1) in

Str.global_replace english_date_format d2 l
with Not_found -> l

let _ = print_string (trans_date ".....08.19..99.....01.27.09............\n")

# let _ = print_string (trans_date ".....08.19..99.....01.27.09............") ;;
.....08.19..99.....27/01/09............
- : unit = ()

a ferramenta ocamllex

sintaxe

um ficheiro ocamllex tem por sufixo .mll e tem a forma seguinte

{
... código OCaml arbitrário ...

}
rule f1 = parse
| regexp1 { acção1 }
| regexp2 { acção2 }
| ...
and f2 = parse

...
and fn = parse

...
{

... código OCaml arbitrário ...
}

a ferramenta ocamllex

compila-se o ficheiro lexer.mll com ocamllex

% ocamllex lexer.mll

tem por efeito a produção de um ficheiro OCaml lexer.ml que contém a
definição de uma função para cada analizador f1, . . . , fn :

val f1 : Lexing.lexbuf -> type1
val f2 : Lexing.lexbuf -> type2
...
val fn : Lexing.lexbuf -> typen

o tipo Lexing.lexbuf

o tipo Lexing.lexbuf é o tipo da estrutura de dados que contém o estado
dum analizador léxico

o módulo Lexing da biblioteca standard fornece vários meios para
construir um valor deste tipo

val from_channel : Pervasives.in_channel -> lexbuf

val from_string : string -> lexbuf

val from_function : (string -> int -> int) -> lexbuf

as expressões regulares em ocamllex

_ (underscore) qualquer caractere
’a’ o caractere ’a’
"foobar" a string "foobar" (em particular ε = )
[caracteres] conjunto de caracteres (por ex. [’a’-’z’ ’A’-’Z’])
[ˆcaracteres] o complemento (por ex. [ˆ ’"’] - tudo menos...)

r1 | r2 a alternativa
r1 r2 a concatenação
r * a estrela/fecho de Kleene
r + uma ou várias repetições de r (def

= r r?)
r ? uma ou zero ocorrências de r (def

= ε | r)

eof fim da entrada

exemplos

identificadores

| [’a’-’z’ ’A’-’Z’] [’a’-’z’ ’A’-’Z’ ’_’ ’0’-’9’]* { ... }

constantes inteiras

| [’0’-’9’]+ { ...}

constantes flutuantes

| [’0’-’9’]+
( ’.’ [’0’-’9’]*
| (’.’ [’0’-’9’]*)? [’e’ ’E’] [’+’ ’-’]? [’0’-’9’]+ )

{ ... }

atalhos

podemos definir atalhos para expressões regulares

let letter = [’a’-’z’ ’A’-’Z’]
let digit = [’0’-’9’]
let decimals = ’.’ digit*
let exponent = [’e’ ’E’] [’+’ ’-’]? digit+

rule token = parse
| letter (letter | digit | ’_’)* { ... }

| digit+ { ... }

| digit+ (decimals | decimals? exponent) { ... }

ambiguidades

para analizadores definidos pela palavra chaves parse, aplica-se a regra da
escolha do lexema que consome o maior texto da entrada

em caso, mesmo assim, de empate, é dada prioridade à regra que aparece
primeiro na definição do analizador

| "fun" { print_endline "função”}
| [’a’-’z’]+ { print_endline ("identificador: ") }

para escolher a regra contrária (o menor consumo da entrada), basta
utilizar a palavra chave shortest no lugar de parse

rule scan = shortest
| regexp1 { acção1 }
| regexp2 { acção2 }
...

reaver o lexema

mecanismo original (presente desde as primeiras versões )

| [’a’-’z’]+ {print_endline (Lexing.lexeme lexbuf) }

em alternativa,

podemos dar um nome à string reconhecida, ou a sub-strings que correspondem a
sub-expressões regulares, com a ajuda da construção as

| [’a’-’z’]+ as s { print_endline s }

| ([’+’ ’-’]? as sign) ([’0’-’9’]+ as n) {print_endline (sign^n)}

processamento dos caracteres brancos

numa acção, é possível invocar recursivamente o analizador léxico, ou um
qualquer outro analizador léxico ali definido

o buffer da analise léxica deve ser passado em argumento ;
está contido - como já o vimos em exemplos anteriores - numa variável
com o nome lexbuf

é assim muito fácil e cómodo tratar dos caracteres brancos :

rule token = parse
| [’ ’ ’\t’ ’\n’]+ { token lexbuf }
| ...

comentários

para tratar dos comentários, podemos utilizar uma expressão regular

... ou um analizador léxico dedicado :

rule token = parse
| "(*" { comment lexbuf }
| ...

and comment = parse
| "*)" { token lexbuf }
| _ { comment lexbuf }
| eof { failwith "comentário inacabado" }

vantagem : tratamos de forma correcta o erro relacionado com um
comentário inacabado

SMDS TC 30



comentários aninhados

outra vantagem : tratamos de forma simples os comentários aninhados

com um contador

rule token = parse
| "(*" { level := 1; comment lexbuf; token lexbuf }
| ...

and comment = parse
| "*)" { decr level; if !level > 0 then comment lexbuf }
| "(*" { incr level; comment lexbuf }
| _ { comment lexbuf }
| eof { failwith "comentário inacabado" }

comentários aninhados

... ou até mesmo sem contador !

rule token = parse
| "(*" { comment lexbuf; token lexbuf }
| ...

and comment = parse
| "*)" { () }
| "(*" { comment lexbuf; comment lexbuf }
| _ { comment lexbuf }
| eof { failwith "comentário inacabado" }

nota : percebemos com este exemplo que ultrapassamos o poder das
expressões regulares com esta construção

um exemplo completo

dado o tipo OCaml para os lexemas

type token =
| Tident of string
| Tconst of int
| Tfun
| Tarrow
| Tplus
| Teof

um exemplo completo

rule token = parse
| [’ ’ ’\t’ ’\n’]+ { token lexbuf }
| "(*" { comment lexbuf }
| "fun" { Tfun }
| [’a’-’z’]+ as s { Tident s }
| [’0’-’9’]+ as s { Tconst (int_of_string s) }
| "+" { Tplus }
| ->" { Tarrow }
| _ as c { failwith ("caractere inválido : " ^

String.make 1 c) }
| eof { Teof }

and comment = parse
| "*)" { token lexbuf }
| _ { comment lexbuf }
| eof { failwith "comentário inacabado" }

localização dos erros

ocamllex propõe um conjunto mínimo de mecanismos para auxiliar o programador na
gestão dos locais onde ocorrem eventuais erros

a estrutura lexbuf dispõe do campo

type lexbuf = { ... mutable lex_curr_p : position;...}

de tipo position

type position = {
(* nome do ficheiro analizado *)
pos_fname : string;
(*contador das linhas - número da linha corrente*)
pos_lnum : int;
(* posição absoluta do inicio da linha corrente *)
pos_bol : int;
(* posição absoluta da posição corrente, em #caracteres *)
pos_cnum : int; }

por omissão, ocamllex só trata da atualização do campo pos_cnum

localização dos erros
é assim necessário inicializar os restantes campos e actualiza-los devidamente quando
necessário

(* função que faz reset às posições no lx (de tipo lexbuf) *)
(* com base no nome do ficheiro f *)
let init_pos lx f = let pos = lx.lex_curr_p in

lx.lex_curr_p <-
{pos with pos_fname = f ; pos_lnum = 0 ; pos_bol = pos.pos_cnum}

para a actualização podemo-nos auxiliar da função new_line (do módulo Lexing)

rule token = parse
...

| [’ ’ ’\t’]+ { token lexbuf }
| ’\n’ { Lexing.newline lexbuf ; token lexbuf }

...
| _ as c { let pos = lexbuf.lex_curr_p in

let line = string_of_int (pos.pos_lnum) in
let col = string_of_int (pos.pos_lnum - pos.pos_bol) in

failwith ("caractere inválido : " ^(String.make 1c)^
" linha : "^line^" coluna : "^col)}

...

as quatro regras

quatro « regras » que não podemos esquecer quando escrevemos um
analizador léxico

1. tratar dos caracteres brancos
2. as regras prioritárias em primeiro (por ex. palavras chaves antes dos

identificadores)
3. assinalar os erros léxicos (caracteres inválidos, mas também

comentários ou strings mal fechados, sequências de escape inválidas,
etc.)

4. tratar do fim de entrada (eof)

eficiência

por omissão ocamllex codifica o autómato numa tabela, que é
interpretada à execução

a opção -ml permite a produção de código OCaml puro, ondeo autómato é
implementado por funções ; esta solução nao é recomendada na prática

eficiência

mesmo utilizando uma tabela, o autómato pode ter uma tamanho
demasiado grande, em particular se a linguagem contempla um número
grande de palavras chaves

é preferível usar uma só expressão regular para capturar os identificadores e
as palavras chaves, e separá-las dentro da acção correspondente com
recurso a uma tabela de palavras-chaves

{
let keywords = Hashtbl.create 97
let () = List.iter (fun (s,t) -> Hashtbl.add keywords s t)

["and", AND; "as", AS; "assert", ASSERT;
"begin", BEGIN; ...

}
rule token = parse
| ident as s

{ try Hashtbl.find keywords s with Not_found -> IDENT s }

case (in)sentitive

se desejamos que um analizador não seja case-sensitive então
evitar absolutamente

| ("a"|"A") ("n"|"N") ("d"|"D")
{ AND }

| ("a"|"A") ("s"|"S")
{ AS }

| ("a"|"A") ("s"|"S") ("s"|"S") ("e"|"E") ("r"|"R") ("t"|"T")
{ ASSERT }

| ...

preferir

rule token = parse
| ident as s

{ let s = String.lowercase s in
try Hashtbl.find keywords s with Not_found -> IDENT s }

compilação e dependência

para compilar (ou recompilar) os módulos Ocaml, é preciso estabelecer as
dependências entre estes módulos, com recurso a ocamldep

contudo, ocamldep não conhece a sintaxe ocamllex ⇒ devemos assegurar
da geração prévia do código ocaml com ocamllex

o Makefile pode ser algo como :

lexer.ml: lexer.mll
ocamllex lexer.mll

.depend: lexer.ml
ocamldep *.ml *.mli > .depend

include .depend

alternativa : usar ocamlbuild

aplicações de ocamllex
(para além da análise léxica)

aplicações de ocamllex

a utilização de ocamllex não está limitada a análise léxica

quando se pretende analisar um texto (string, ficheiro, fluxo, sequência de
pacotes formatados de informação, etc.) com base em expressões regulares,
ocamllex é uma ferramenta particularmente adaptada

em particular para escrever filtros, i.e. programas que traduzem uma
linguagem para uma outra via modificações locais e relativamente simples

exemplo 1

exemplo 1 : agrupar várias linhas em branco consecutivas numa só

tão simples como

rule scan = parse
| ’\n’ ’\n’+ { print_string "\n\n"; scan lexbuf }
| _ as c { print_char c; scan lexbuf }
| eof { () }

{ let () = scan (Lexing.from_channel stdin) }

o processo de compilação pode ser o seguinte

% ocamllex mbl.mll
% ocamlopt -o mbl mbl.ml

a utilização segue o padrão seguinte

% ./mbl < infile > outfile

exemplo 2

exemplo 2 : contar as ocorrências de uma determinada palavra num texto

{
let word = Sys.argv.(1)
let count = ref 0

}
rule scan = parse
| [’a’-’z’ ’A’-’Z’]+ as w

{ if word = w then incr count; scan lexbuf }
| _

{ scan lexbuf }
| eof

{ () }
{
let () = scan (Lexing.from_channel (open_in Sys.argv.(2)))
let () = Printf.printf "%d occurrence(s)\n" !count

}

exemplo 3 : caml2html

exemplo 3 : um pequeno tradutor de Ocaml para HTML,
para embelezar código fonte numa visualização web/online

objetivo
• uso : caml2html file.ml, que produz file.ml.html
• palavras chaves a verde, comentários a vermelho
• numerar as linhas
• isto tudo em menos de 100 linhas de código

caml2html

escrevemos tudo num só ficheiro caml2html.mll

começamos por verificar as opções da linha de comando

{
let () =

if Array.length Sys.argv <> 2
|| not (Sys.file_exists Sys.argv.(1)) then begin

Printf.eprintf "usage: caml2html file\n";
exit 1

end

em seguida abrimos o ficheiro HTML em modo escrita e o actualizamos
com recurso à função fprintf

let file = Sys.argv.(1)
let cout = open_out (file ^ ".html")
let print s = Printf.fprintf cout s

caml2html

escrevemos o início do ficheiro HTML definindo como título o nome do
ficheiro
utilizamos a tag HTML <pre> para formatar o código de acordo com a
formatação original

let () =
print «html><head><title>%s</title></head><body>\n<pre>" file

introduzimos uma função para numerar cada linha,
e invocamo-la imediatamente para a primeira linha

let count = ref 0
let newline () = incr count; print "\n%3d: " !count
let () = newline ()

caml2html

definimos uma tabela de palavras-chaves (como no caso da análise léxica)

let is_keyword =
let ht = Hashtbl.create 97 in
List.iter

(fun s -> Hashtbl.add ht s ())
[ "and"; "as"; "assert"; "asr"; "begin"; "class";

... ];
fun s -> Hashtbl.mem ht s

}

introduzimos uma expressão regular para os identificadores

let ident =
[’A’-’Z’ ’a’-’z’ ’_’] [’A’-’Z’ ’a’-’z’ ’0’-’9’ ’_’]*

caml2html

podemos agora tratar do analizador em si

para um identificador, testamos se se trata de uma palavra-chave

rule scan = parse
| ident as s

{ if is_keyword s then begin
print «font color=\"green\»%s</font>" s

end else
print "%s" s;

scan lexbuf }

a cada fim de linha, invocamos a função newline :

| "\n"
{ newline (); scan lexbuf }

caml2html

para um comentário, mudamos a cor (vermelho) e invocamos outro
analizador comment especializado na análise de comentários ; aquando do
seu retorno, retomamos a cor por omissão e a análise por scan

| "(*"
{ print «font color=\"990000\»(*";

comment lexbuf;
print «/font>";
scan lexbuf }

qualquer outro caractere é impresso tal e qual

| _ as s { print "%s" s; scan lexbuf }

presenciando o fim da entrada, terminamos

| eof { () }

caml2html

para a análise dos comentários, basta não esquecer o processamento de
newline :

and comment = parse
| "(*" { print "(*"; comment lexbuf; comment lexbuf }
| "*)" { print "*)" }
| eof { () }
| "\n" { newline (); comment lexbuf }
| _ as c { print "%c" c; comment lexbuf }

termina-se com a aplicação do analizador scan sobre o ficheiro de entrada

{
let () =

scan (Lexing.from_channel (open_in file));
print «/pre>\n</body></html>\n";
close_out cout

}

caml2html

está quase certo... :

• o caractere < tem significado em HTML, a sua tradução de OCaml
deve assim ser &lt;
• da mesma forma é preciso traduzir & para &amp;

• uma string OCaml pode conter (*
(como é precisamente o caso neste programa !)

caml2html

corrijamos !

juntamos uma regra para < e um analizador para as strings

| “<" { print "&lt;"; scan lexbuf }
| "&" { print "&amp;"; scan lexbuf }
| ’"’ { print "\; string lexbuf; scan lexbuf }

é preciso ter em atenção ao caractere ’"’, quando " não indica o início de
uma string

| "’\"’"
| _ as s { print "%s" s; scan lexbuf }

caml2html

reflectimos este tratamento ao processo dos comentários (de facto, em
OCaml podemos comentar código contendo "*)")

| ’"’ { print "\; string lexbuf; comment lexbuf }
| "’\"’"
| _ as s { print "%s" s; comment lexbuf }

finalmente, as strings são tratadas pelo analizador string, sem nos
esquecermos das sequências de escape (tais como \" por exemplo)

and string = parse
| ’"’ { print "\ }
| «" { print "&lt;"; string lexbuf }
| "&" { print "&amp;"; string lexbuf }
| "\\" _
| _ as s { print "%s" s; string lexbuf }

caml2html

agora sim, funciona como pretendido
(um bom teste é experimentar sobre o próprio caml2html.mll)

para ser perfeito, deveríamos também converter as tabulações em início de
linha (tipicamente inseridos pelos editores de texto) como espaços

a alteração é deixada como exercício...

exemplo 4

exemplo 4 : indentação automática de programas C

ideia :
• em cada abertura de chaveta, aumentamos a tabulação
• em cada fecho de chaveta, diminuímo-la
• em cada fim de linha, imprimimos a tabulação actual
• sem esquecer o processamento das strings e dos comentários

SMDS TC 57



indentação automática dos programas C

dados os meios de manter o estado das tabulações e das respectivas
visualizações

{
open Printf

let margin = ref 0
let print_margin () =

printf "\n"; for i = 1 to 2 * !margin do printf " " done
}

SMDS TC 58



indentação automática dos programas C

em cada fim de linha, ignoramos os espaços em início de linha e mostramos
a tabulação actual

let space = [’ ’ ’\t’]

rule scan = parse
| ’\n’ space*

{ print_margin (); scan lexbuf }

as chavetas modificam a tabulação corrente

| "{"
{ incr margin; printf "{"; scan lexbuf }

| "}"
{ decr margin; printf "}"; scan lexbuf }

indentação automática dos programas C

casos particulares : um fecho de chaveta em início de linha deve diminuir a
tabulação antes da sua produção

| ’\n’ space* "}"
{ decr margin; print_margin (); printf "}";

scan lexbuf }

não esqueçamos as strings

| ’"’ ([^ ’\\’ ’"’] | ’\\’ _)* ’"’ as s
{ printf "%s" s; scan lexbuf }

nem os comentários numa linha

| "//" [^ ’\n’]* as s
{ printf "%s" s; scan lexbuf }

indentação automática dos programas C

nem os comentários da forma /* ... */

| "/*"
{ printf "/*"; comment lexbuf; scan lexbuf }

onde

and comment = parse
| "*/"

{ printf "*/" }
| eof

{ () }
| _ as c

{ printf "%c" c; comment lexbuf }

indentação automática dos programas C

para terminar, qualquer outra situação é reproduzida tal e qual

eof termina a análise

rule scan = parse
| ...
| _ as c

{ printf "%c" c; scan lexbuf }
| eof

{ () }

o programa principal cabe em duas linhas

{
let c = open_in Sys.argv.(1)
let () = scan (Lexing.from_channel c); close_in c

}

indentação automática dos programas C

nesta forma, ainda há obviamente margem para melhorias

em particular, um corpo de um ciclo ou de uma condicional está reduzida a
uma só instrução não sofrerá indentação suplementar :

if (x==1)
continue;

exercício : replicar este exercício para OCaml (é bastante mais difícil, mas
poderá limitar-se a alguns casos simples)

análise sintáctica

não é só o léxico que importa, a sintaxe e a morfologia da frase também

Man eating piranha mistakenly sold as pet fish

Babies are what the mother eats

análise sintáctica

o objectivos da análise sintáctica é reconhecer as frases pertencendo à
sintaxe da linguagem

o que o fluxo de caracteres é para a análise léxica, o fluxo dos lexemas –
resultado desta mesma análise – o são para a análise sintáctica

a saída da análise sintáctica é uma árvore de sintaxe abstracta

análise sintáctica

sequência de caracteres
"fun x -> ( x + 1 )"

↓
análise léxica

↓

sequência de lexemas

fun x -> ( x + 1 )

sequência de lexemas

fun x -> ( x + 1 )

↓
análise sintáctica

↓

sintaxe abstracta
Fun

"x" App

App

Op

+

Var

"x"

Const

1

OCamlyacc

yacc, em prática

nesta aula vamos dar mais ênfase à ferramenta menhir

a introdução ao ocamlyacc far-se-á em consequência por uma explicação
prática dos problemas encontrados na engenharia das gramáticas
LALR(1) (que se estendem naturalmente às gramáticas LR(1), mais
permissivas no que diz respeito ao risco de conflito)

todos os exemplos apresentados são compatíveis com o menhir (podemos
usar o menhir no lugar do ocamlyacc sobre os mesmos ficheiros
apresentados)

precedência e associatividade
%{
(* Cabeçalho. Opcional, pode ser omitido. *)
(* Contém eventual código OCaml *)
%}
%token INT PLUS MINUS TIMES DIV EOF
%start expr
%type <unit> expr
%%

expr: INT { (* acção - código OCaml *) }
| expr PLUS expr {}
| expr MINUS expr {}
| expr TIMES expr {}
| expr DIV expr {}
| MINUS expr {};

%%
(* Trailer. Opcional : pode ser omitido.*)
(* Contém eventual código OCaml *)

precedência e associatividade - (ocamlyacc -v calc.mly)

0 $accept : %entry% $end
1 expr : INT
2 | expr PLUS expr
3 | expr MINUS expr
4 | expr TIMES expr
5 | expr DIV expr
6 | MINUS expr
7 %entry% : ’\001’ expr

(.....)
14: shift/reduce conflict (shift 7, reduce 5) on PLUS
14: shift/reduce conflict (shift 8, reduce 5) on MINUS
14: shift/reduce conflict (shift 9, reduce 5) on TIMES
14: shift/reduce conflict (shift 10, reduce 5) on DIV
state 14
expr : expr . PLUS expr (2)
expr : expr . MINUS expr (3)
expr : expr . TIMES expr (4)
expr : expr . DIV expr (5)
expr : expr DIV expr . (5)

PLUS shift 7
MINUS shift 8
TIMES shift 9
DIV shift 10
$end reduce 5
(.....)
State 14 contains 4 shift/reduce conflicts.
9 terminals, 3 nonterminals
8 grammar rules, 15 states

precedência e associatividade: a solução

%{
%}
%token INT PLUS MINUS TIMES DIV EOF
%left PLUS MINUS
%left TIMES DIV
%left uminus
%start expr
%type <unit> expr
%%

expr: INT {}
| expr PLUS expr {}
| expr MINUS expr {}
| expr TIMES expr {}
| expr DIV expr {}
| MINUS expr %prec uminus {};

dangling else

Consideremos as três regras: (1) S → if E then S else S ,
(2) S → if E then S e (3) S → resto

• Em if a then if b then x else y =⇒ a que if pertence o else?

• Conflito shift/reduce com else entre
(reduce) S → if E then S . e (shift) S → if E then S . else S

• uma solução: reescrever a gramática desta forma
1 S → M
2 S → U
3 M → if E then M else M
4 M → resto
5 U → if E then S
6 U → if E then M else U

• outra solução: deixar como está! De facto em caso de conflito shift/reduce,
as ferramentas como yacc assumem a interpretação shift.

• Este truque deve ser usado com cuidado! Em caso de conflito verificar com
cuidado se podem deixar o yacc decidir sempre pelo shift.

em concreto...(ocamlyacc -v exemplo.mly)

%{
%}

%token ID WHILE BEGIN END DO IF THEN ELSE SEMI ASSIGN EOF
%start prog
%type <unit>prog
%%

prog: stmlist EOF {};

stmlist: stm {}
| stmlist SEMI stm {};

stm: ID ASSIGN ID {}
| WHILE ID DO stm {}
| BEGIN stmlist END {}
| IF ID THEN stm {}
| IF ID THEN stm ELSE stm {};

em concreto...(ocamlyacc -v exemplo.mly)
0 $accept : %entry% $end
1 prog : stmlist EOF
2 stmlist : stm
3 | stmlist SEMI stm
4 stm : ID ASSIGN ID
5 | WHILE ID DO stm
6 | BEGIN stmlist END
7 | IF ID THEN stm
8 | IF ID THEN stm ELSE stm
9 %entry% : ’\001’ prog

(......)
22: shift/reduce conflict (shift 23, reduce 7) on ELSE
state 22
stm : IF ID THEN stm . (7)
stm : IF ID THEN stm . ELSE stm (8)

ELSE shift 23
END reduce 7
SEMI reduce 7
EOF reduce 7
(......)
State 22 contains 1 shift/reduce conflict.
14 terminals, 5 nonterminals
10 grammar rules, 25 states

conflitos reduce/reduce

%token ID PLUS MINUS AND OR EQUAL ASSIGN EOF
%left OR
%left AND
%left PLUS
%start stm
%type <unit> stm
%%

stm: ID ASSIGN ae {}
| ID ASSIGN be {};

be: be OR be {}
| be AND be {}
| ae EQUAL ae {}
| ID {};

ae: ae PLUS ae {}
| ID {};

SMDS TC 75



conflitos reduce/reduce
0 $accept : %entry% $end
1 stm : ID ASSIGN ae
2 | ID ASSIGN be
3 be : be OR be
4 | be AND be
5 | ae EQUAL ae
6 | ID
7 ae : ae PLUS ae
8 | ID

(.....)
6: reduce/reduce conflict (reduce 6, reduce 8) on $end
state 6
be : ID . (6)
ae : ID . (8)

$end reduce 6
PLUS reduce 8
AND reduce 6
OR reduce 6
EQUAL reduce 8
(.....)
State 6 contains 1 reduce/reduce conflict.
11 terminals, 5 nonterminals
10 grammar rules, 19 states

conflitos reduce/reduce

aqui a dificuldade é que um identificador pode ocorrer tanto numa
expressão booleana (be) como numa expressão aritmética (ae)

na analise dum identificador, que produção escolher para a redução? a
regra 6 (identificador booleano) ou a regra 8 (identificador aritmético)?

nesta fase da compilação não temos forma fácil de inferir em que situação
nos encontramos. Por isso o mais sensato é reescrever a gramática de tal
forma que a escolha seja feita posteriormente: na análise semântica
(tipagem, análise de porte etc.)

conflitos reduce/reduce

%token ID PLUS MINUS AND OR EQUAL ASSIGN EOF
%left OR
%left AND
%left PLUS
%start stm
%type <unit> stm
%%

stm: ID ASSIGN expr {};

expr: expr OR expr {}
| expr AND expr {}
| expr EQUAL expr {}
| expr PLUS expr {}
| ID {};

gestão implícita dos conflitos pelo yacc
(Documentação do OCamlYacc)

The precedence declarations are used in the following way to resolve reduce/reduce and
shift/reduce conflicts:

• Tokens and rules have precedences. By default, the precedence of a rule is the
precedence of its rightmost terminal. You can override this default by using the
%prec directive in the rule.

• A reduce/reduce conflict is resolved in favor of the first rule (in the order given by
the source file), and ocamlyacc outputs a warning.

• A shift/reduce conflict is resolved by comparing the precedence of the rule to be
reduced with the precedence of the token to be shifted. If the precedence of the
rule is higher, then the rule will be reduced; if the precedence of the token is
higher, then the token will be shifted.

• A shift/reduce conflict between a rule and a token with the same precedence will
be resolved using the associativity: if the token is left-associative, then the parser
will reduce; if the token is right-associative, then the parser will shift. If the token
is non-associative, then the parser will declare a syntax error.

• When a shift/reduce conflict cannot be resolved using the above method, then
ocamlyacc will output a warning and the parser will always shift.

yacc: o token error

o que acontece se o yacc detecta um erro (sintáctico) durante a análise sintáctica?

isto acontece quando a palavra actualmente analisada deixa de poder ter
continuação que permita ser reconhecida

neste caso ferramentas como yacc utilizam a técnica da correcção local do erro

esta técnica baseia-se na introdução dum token especial, o token error , no local
onde o erro foi detectado e na presença de tokens que desempenham um papel de
sincronização

esses tokens são em geral tokens de pontuação que permitam delimitar
componentes importantes da linguagem em causa

o token error pode assim ser utilizado como um token normal na gramática. A
sua utilização num local particular da gramática representa o facto que a
produção em questão pretende tratar um eventual erro sintáctico onde o token
error aparece

yacc: o token error

que faz o yacc com o símbolo error introduzido aquando da detecção
efectiva dum erro?
1. esvaziar a pilha de estados (se necessário) até chegar a um estado no

qual uma acção para o token error seja shift

2. executar shift error

3. descartar símbolos de entrada (se necessário) até atingir um estado
que tenha uma acção normal (que não seja de “erro”) sobre o token de
antevisão (lookahead token). Este token é o token de
sincronização

4. retomar o parsing normalmente.

yacc: o token error

os tokens de sincronização podem ser pontuação que separam instruções

neste caso se um erro ocorrer, vamos procurar o próximo ponto de
sincronização e ignorar todo o input que ocorre entre o local de erro e este
ponto

a análise pode recomeçar normalmente a seguir. I.e. deita fora toda a
instrução onde o erro ocorreu e retoma a analise normalmente com a
instrução seguinte

gestão dos erros pelo Yacc

(Documentação do OCamlYacc)

• Error recovery is supported as follows: when the parser reaches an
error state (no grammar rules can apply), it calls a function named
parse_error with the string "syntax error"as argument. The default
parse_error function does nothing and returns, thus initiating error
recovery (see below). The user can define a customized parse_error
function in the header section of the grammar file.
• The parser also enters error recovery mode if one of the grammar

actions raises the Parsing.Parse_error exception.
• In error recovery mode, the parser discards states from the stack until

it reaches a place where the error token can be shifted. It then
discards tokens from the input until it finds three successive tokens
that can be accepted, and starts processing with the first of these. If
no state can be uncovered where the error token can be shifted, then
the parser aborts by raising the Parsing.Parse_error exception.

a ferramenta menhir

menhir

Menhir é um ferramenta que transforma uma gramática num analisador
OCaml ; este assenta numa análise LR(1)

cada produção da gramática está acompanhada de uma ação semântica
i.e. de código OCaml que constrói um valor semântico (tipicamente uma
árvore de sintaxe abstracta)

Menhir utiliza-se em conjunto com um analisador léxico
(tipicamente ocamllex)

estrutura

um ficheiro Menhir tem o sufixo .mly e tem por estrutura

%{
...(opcional) código OCaml arbitrário ...

%}
...declaração dos tokens...
...declaração de precedências e associatividade...
...declaração dos pontos de entrada...
%%
não-terminal-1:
| produção { ação }
| produção { ação }
;

não-terminal-2:
| produção { ação } ;
...
%%
... (opcional) código OCaml arbitrário ...

SMDS TC 86



exemplo minimalista

%token PLUS LPAR RPAR EOF
%token <int> INT
%start <int> frase
%%

frase:
e = expression; EOF { e }

;

expression:
| e1 = expression; PLUS; e2 = expression { e1 + e2 }
| LPAR; e = expression; RPAR { e }
| i = INT { i }
;

exemplo

compilamos o ficheiro arith.mly da forma seguinte

% menhir -v arith.mly

obtemos código OCaml puro no ficheiro arith.ml(i), que contem em
particular

• a declaração de um tipo token

type token = RPAR | PLUS | LPAR | INT of int | EOF

• para cada não-terminal declarado com %start, uma função de tipo

val frase: (Lexing.lexbuf -> token) -> Lexing.lexbuf -> int

como vemos, esta função toma como parâmetro um analisador léxico,
do tipo dos que ocamllex produz (cf aula sobre análise léxica)

ocamllex + menhir

quando combinamos ocamllex e menhir

• lexer.mll faz referência aos lexemas definidos em parser.mly

{
open Parser

}
...

• o analisador léxico e o analisador sintáctico são combinados da forma
seguinte:

let c = open_in file in
let lb = Lexing.from_file c in
let e = Parser.frase Lexer.token lb in
...

ocamllex + menhir

lexer.mll parser.mly

lexer.ml parser.mli parser.ml

main.ml

ocamllex menhir

= fonte utilizador
= construido automaticamente = dependência

conflitos

revisitemos a noção de conflito, mas desta vez com o Menhir

quando a gramática não é LR(1), Menhir apresenta os conflitos ao
utilizador

• o ficheiro .automaton contém uma descrição do autómato LR(1) ; os
conflito estão ali referenciados

• o ficheiro .conflicts contém, se assim for o caso, uma explicação de
cada conflito, na forma de uma sequência de lexemas que conduz a
duas árvores de derivação (i.e. a uma manifestação do conflito)

exemplo

na gramática anterior, Menhir assinala o conflito

% menhir -v arith.mly
Warning: one state has shift/reduce conflicts.
Warning: one shift/reduce conflict was arbitrarily resolved.

o ficheiro arith.automaton contém em particular

State 6:
expression -> expression . PLUS expression [ RPAR PLUS EOF ]
expression -> expression PLUS expression . [ RPAR PLUS EOF ]
– On PLUS shift to state 5
– On RPAR reduce production expression -> expression PLUS expression
– On PLUS reduce production expression -> expression PLUS expression
– On EOF reduce production expression -> expression PLUS expression
** Conflict on PLUS

exemplo

o ficheiro arith.conflicts apresenta a seguinte explicação do conflito

** Conflict (shift/reduce) in state 6.
** Token involved: PLUS
** This state is reached from phrase after reading:

expression PLUS expression

** In state 6, looking ahead at PLUS, shifting is permitted
** because of the following sub-derivation:

expression PLUS expression
expression . PLUS expression

** In state 6, looking ahead at PLUS, reducing production
** expression -> expression PLUS expression
** is permitted because of the following sub-derivation:

expression PLUS expression // lookahead token appears

resolução dos conflitos

uma forma de resolução dos conflitos passa por indicar ao Menhir como
escolher entre leituras e reduções

para esse efeito, podemos dar prioridades aos lexemas e às produções
e regras de associatividades

como no caso do yacc

por omissão, a prioridades de uma produção é a prioridade do seu lexema
mais a direita (mas esta pode ser especificada de forma explícita)

resolução de conflitos

se a prioridade da produção é maior do que o lexema por ler,
então favorece-se a redução

reciprocamente, sa prioridade do lexema for maior,
então a leitura é privilegiada

em caso de igualdade, consulta-se as regras de associatividade : um lexema
associativo à esquerda favorece a redução e um lexema associativo à direita
favorece a leitura

exemplo

no nosso exemplo, basta indicar por exemplo que PLUS é associativo à
esquerda

%token PLUS LPAR RPAR EOF
%token <int> INT
%left PLUS
%start <int> frase
%%
frase:

e = expression; EOF { e }
;
expression:
| e1 = expression; PLUS; e2 = expression { e1 + e2 }
| LPAR; e = expression; RPAR { e }
| i = INT { i }
;

prioridades

para associar prioridades aos lexemas, utilizamos a convenção seguinte :
• a ordem de declaração da associatividade fixa as prioridades

(os primeiros lexemas têm as prioridades mais fracas)
• vários lexemas podem aparecer numa mesma declaração (na mesma

linha),
tendo assim a mesma prioridade

exemplo :

%left PLUS MINUS
%left TIMES DIV

um clássico...

a gramática contém um conflito

expression:
| IF e1 = expression; THEN; e2 = expression

{ ... }
| IF e1 = expression; THEN; e2 = expression;

ELSE; e3 = expression
{ ... }

| i = INT
{ ... }

| ...

um clássico...

corresponde à situação

IF a THEN IF b THEN c ELSE d

para associar o ELSE ao THEN mais próximo, devemos privilegiar a leitura

%nonassoc THEN
%nonassoc ELSE

(esta situação é conhecida em inglês como dangling else)

as vantagens do Menhir

Menhir oferece numerosas vantagens relativamente a ferramentas como,
por exemplo, ocamlyacc :
• não-terminais parametrizados por (não-)terminais

• em particular, mecanismos para escrever expressões regulares na
gramática (E?, E*, E+), listas com separadores

expr: (...);
lista_exprs: ve = separated_list(COMMA,expr) { ve } ;

• explicações dos conflitos
• modo interactivo, modo depuração
• gramáticas parametrizadas
• análise LR(1) no lugar de LALR(1)
• etc.

ler o manual de Menhir !

localizações

para que as fases seguintes da análise sintáctica (e.g. a tipagem) possam
dar indicações de posição nas mensagens de erro, convém guardar
informação de localização directamente na árvore de sintaxe abstracta

Menhir fornece esta informação via $startpos e $endpos, dois valores de
tipo Lexing.position ; esta informação é-lhe transmitida pelo analisador
léxico

cuidado : relembramos que ocamllex só mantém por omissão a posição
absoluta dentro do ficheiro ; ter os números de linha e coluna actualizados
necessita de um tratamento explícito realizado pelo programador (quem
define os analisadores), (ver por exemplo, os vários ficheiros lexer.mll
fornecidos nas aulas práticas)

localizações

uma forma de conservar a informação de localização na AST é a seguinte
(ver aula sobre tipagem)

type expression =
{ desc: desc;

loc : Lexing.position * Lexing.position }

and desc =
| Econst of int
| Eplus of expression * expression
| Eneg of expression
| ...

localização

a gramática pode então parecer-se com

expression:
| d = desc { { desc = d; loc = $startpos, $endpos } }
;

desc:
| i = INT { Econst i }
| e1 = expression; PLUS; e2 = expression { Eplus (e1, e2) }
| ...

compilação e dependências

como no caso de ocamllex, é necessário assegurar a aplicação de menhir
antes do cálculo das dependências

o Makefile pode ser algo como :

lexer.ml: lexer.mll
ocamllex lexer.mll

parser.mli parser.ml: parser.mly
menhir -v parser.mly

.depend: lexer.ml parser.mli parser.ml
ocamldep *.ml *.mli > .depend

include .depend

lexer.mll parser.mly

lexer.ml parser.mli parser.ml

main.ml

ocamllex menhir

... ou então, utilizar ocamlbuild

conclusão

leituras de referência

estes acetatos resultam essencialmente de uma adaptação do material pedagógico
gentilmente cedido pelo Jean-Christophe Filliâtre (link1, link2)

adicionalmente poderá consultar as obras seguintes

• Modern Compilers: Principles, Techniques, and
Tools, Alfred V. Aho, Monica S. Lam, Ravi Sethi et
Jeffrey D. Ullman

• Types and Programming Languages, Benjamin
C. Pierce

• Modern Compiler Implementation, Andrew W.
Appel (3 versões: ML, C, Java)

